

NAG Library Routine Document

F07FJF (DPOTRI)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07FJF (DPOTRI) computes the inverse of a real symmetric positive definite matrix A , where A has been factorized by F07FDF (DPOTRF).

2 Specification

```
SUBROUTINE F07FJF (UPLO, N, A, LDA, INFO)
INTEGER           N, LDA, INFO
REAL (KIND=nag_wp) A(LDA,*)
CHARACTER(1)      UPLO
```

The routine may be called by its LAPACK name *dpotri*.

3 Description

F07FJF (DPOTRI) is used to compute the inverse of a real symmetric positive definite matrix A , the routine must be preceded by a call to F07FDF (DPOTRF), which computes the Cholesky factorization of A .

If $\text{UPLO} = \text{'U}'$, $A = U^T U$ and A^{-1} is computed by first inverting U and then forming $(U^{-1})U^{-T}$.

If $\text{UPLO} = \text{'L}'$, $A = LL^T$ and A^{-1} is computed by first inverting L and then forming $L^{-T}(L^{-1})$.

4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

5 Arguments

1: UPLO – CHARACTER(1)	<i>Input</i>
<i>On entry:</i> specifies how A has been factorized.	
UPLO = 'U'	
$A = U^T U$, where U is upper triangular.	
UPLO = 'L'	
$A = LL^T$, where L is lower triangular.	
<i>Constraint:</i> UPLO = 'U' or 'L'.	
2: N – INTEGER	<i>Input</i>
<i>On entry:</i> n , the order of the matrix A .	
<i>Constraint:</i> $N \geq 0$.	

3:	$A(\text{LDA}, *)$ – REAL (KIND=nag_wp) array	<i>Input/Output</i>
Note: the second dimension of the array A must be at least $\max(1, N)$.		
<i>On entry:</i> the upper triangular matrix U if $\text{UPLO} = \text{'U'}$ or the lower triangular matrix L if $\text{UPLO} = \text{'L'}$, as returned by F07FDF (DPOTRF).		
<i>On exit:</i> U is overwritten by the upper triangle of A^{-1} if $\text{UPLO} = \text{'U'}$; L is overwritten by the lower triangle of A^{-1} if $\text{UPLO} = \text{'L'}$.		
4:	LDA – INTEGER	<i>Input</i>
<i>On entry:</i> the first dimension of the array A as declared in the (sub)program from which F07FJF (DPOTRI) is called.		
<i>Constraint:</i> $\text{LDA} \geq \max(1, N)$.		
5:	INFO – INTEGER	<i>Output</i>
<i>On exit:</i> $\text{INFO} = 0$ unless the routine detects an error (see Section 6).		

6 Error Indicators and Warnings

$\text{INFO} < 0$

If $\text{INFO} = -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

$\text{INFO} > 0$

Diagonal element $\langle \text{value} \rangle$ of the Cholesky factor is zero; the Cholesky factor is singular and the inverse of A cannot be computed.

7 Accuracy

The computed inverse X satisfies

$$\|XA - I\|_2 \leq c(n)\epsilon\kappa_2(A) \quad \text{and} \quad \|AX - I\|_2 \leq c(n)\epsilon\kappa_2(A),$$

where $c(n)$ is a modest function of n , ϵ is the **machine precision** and $\kappa_2(A)$ is the condition number of A defined by

$$\kappa_2(A) = \|A\|_2\|A^{-1}\|_2.$$

8 Parallelism and Performance

F07FJF (DPOTRI) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately $\frac{2}{3}n^3$.

The complex analogue of this routine is F07FWF (ZPOTRI).

10 Example

This example computes the inverse of the matrix A , where

$$A = \begin{pmatrix} 4.16 & -3.12 & 0.56 & -0.10 \\ -3.12 & 5.03 & -0.83 & 1.18 \\ 0.56 & -0.83 & 0.76 & 0.34 \\ -0.10 & 1.18 & 0.34 & 1.18 \end{pmatrix}.$$

Here A is symmetric positive definite and must first be factorized by F07FDF (DPOTRF).

10.1 Program Text

```
Program f07fjfe

!     F07FJF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
Use nag_library, Only: dpotrf, dpotri, nag_wp, x04caf
!     .. Implicit None Statement ..
Implicit None
!     .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6
!     .. Local Scalars ..
Integer :: i, ifail, info, lda, n
Character (1) :: uplo
!     .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:,:)
!     .. Executable Statements ..
Write (nout,*) 'F07FJF Example Program Results'
Skip heading in data file
Read (nin,*)
Read (nin,*) n
lda = n
Allocate (a(lda,n))

!     Read A from data file

Read (nin,*) uplo
If (uplo=='U') Then
    Read (nin,*)(a(i,i:n),i=1,n)
Else If (uplo=='L') Then
    Read (nin,*)(a(i,1:i),i=1,n)
End If

!     Factorize A
!     The NAG name equivalent of dpotrf is f07fdf
Call dpotrf(uplo,n,a,lda,info)

Write (nout,*)
Flush (nout)
If (info==0) Then

    !     Compute inverse of A
    !     The NAG name equivalent of dpotri is f07fjf
    Call dpotri(uplo,n,a,lda,info)

!     Print inverse

!     ifail: behaviour on error exit
!           =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
    ifail = 0
    Call x04caf(uplo,'Nonunit',n,n,a,lda,'Inverse',ifail)
```

```

Else
  Write (nout,*), 'A is not positive definite'
End If

End Program f07fjfe

```

10.2 Program Data

```

F07FJF Example Program Data
 4 :Value of N
'L' :Value of UPLO
4.16
-3.12 5.03
 0.56 -0.83 0.76
-0.10 1.18 0.34 1.18 :End of matrix A

```

10.3 Program Results

F07FJF Example Program Results

Inverse	1	2	3	4
1	0.6995			
2	0.7769	1.4239		
3	0.7508	1.8255	4.0688	
4	-0.9340	-1.8841	-2.9342	3.4978
